

『例題で学ぶ Excel VBA 入門』

演習課題・解答例

1 はじめての VBA

【課題 1.1】

```
Sub 文字列のボックス出力()  
    MsgBox "自分の姓名は" & "鈴木" & "次郎"  
End Sub
```

【課題 1.2】

```
Sub 関数のボックス出力()  
    MsgBox "√5=" & Sqr(5)  
End Sub
```

【課題 1.3】

```
Sub セルからセルへ()  
    Sheets("Sheet1").Select  
    Range("C2").Value = Range("B2").Value  
End Sub
```

【課題 1.4】

```
Sub 桁数()  
    MsgBox "1234567 の桁数は" & Int(Log(1234567) / Log(10)) + 1  
End Sub
```

2 変数の活用

【課題 2.1】

```
Sub 占い()  
    Dim anata As Integer, oaite As Integer, ans As Integer  
    anata = Val(InputBox("あなたの年齢は?"))  
    oaite = Val(InputBox("お相手の年齢は?"))  
    ans = (anata * oaite) Mod 100  
    MsgBox "恋が実る確率は" & ans & "%です."  
End Sub
```

【課題 2.2】

```
Sub ID から学番()  
    Dim ID As String, gakuban As String  
    ID = InputBox("パソコンユーザ ID は?")  
    gakuban = UCase(Mid(ID, 2, 7))  
    MsgBox "学籍番号は" & gakuban  
End Sub
```

【課題 2.3】

```
Sub 学番から ID()  
    Dim gakuban As String, myouji As String, ID As String  
    gakuban = InputBox("学籍番号は?")  
    myouji = InputBox("苗字 (ローマ字) は?")  
    ID = LCase("s" & gakuban & Left(myouji, 1))  
    MsgBox "パソコンユーザ ID は" & ID  
End Sub
```

3 セル・セル範囲の操作

【課題 3.1】

```
Sub 範囲内フォント黒に()  
    Dim rng As Range  
    MsgBox "B2:B4 のフォントを黒に"  
    Sheets("Sheet1").Select  
    Set rng = Range("B2:B4")  
    rng.Font.ColorIndex = 1  
End Sub
```

【課題 3.2】

```
Sub セル範囲の行列を入れ替えて貼り付け()  
    MsgBox "D2:D6 を F2 に行列を入れ替えて貼り付け"  
    Range("D2:D6").Copy  
    Range("F2").PasteSpecial Transpose:=True  
End Sub
```

4 条件分岐

【課題 4.1】

```
Sub 運転免許 2()  
    Dim age As Integer  
    age = Val(InputBox("あなたの年齢は?"))  
    If age >= 18 Then  
        MsgBox "自動車運転免許取得可能です。"  
    ElseIf age >= 16 Then  
        MsgBox "原付運転免許取得可能です。"  
    Else  
        MsgBox "どの運転免許もまだです。"  
    End If  
End Sub
```

【課題 4.2】

```
Sub 内閣支持率()  
    Dim ritu As Double  
    ritu = Val(InputBox("内閣支持率は?"))  
    If ritu > 100 Then  
        MsgBox "100%を超える支持率はありません。"  
    ElseIf ritu < 0 Then  
        MsgBox "0%未満の支持率はありません。"  
    ElseIf ritu < 30 Then  
        MsgBox "危険水域です。"  
    Else  
        MsgBox "まだ大丈夫でしょう。"  
    End If  
End Sub
```

【課題 4.3】

```
Sub 商学部判定()  
    Dim gakuban As String  
    gakuban = InputBox("学籍番号かパソコンユーザ ID を入力")  
    If Left(gakuban, 1) = "s" Then  
        gakuban = UCase(Mid(gakuban, 2, 7))  
    End If  
    If Mid(gakuban, 4, 1) = "C" Then  
        MsgBox "商学部の学生です。"  
    Else  
        MsgBox "商学部の学生ではありません。"  
    End If  
End Sub
```

【課題 4.4】

```
Sub 運転免許番外編()  
    Dim n As Integer, s As String, h As String  
    Sheets("Sheet1").Select  
    n = Range("B1").Value  
    s = Range("B2").Value  
    If n >= 18 Then  
        h = "自動車運転免許取得可能"  
    Else  
        h = "自動車運転免許取得不可"  
    End If  
    If s = "男" Then  
        MsgBox h & "だぜ!"  
    Else  
        MsgBox h & "だわ!"  
    End If  
End Sub
```

【課題 4.5】

```
Sub プレナベアツ()  
    Dim i As Integer, sakebi As String  
    i = Val(InputBox("1 以上 40 以下の整数を入力"))  
    sakebi = ""  
    If i Mod 3 = 0 Then  
        sakebi = "あへつ。"  
    ElseIf i Mod 10 = 3 Then  
        sakebi = "あへつ。"  
    ElseIf i ¥ 10 = 3 Then  
        sakebi = "あへつ。"  
    End If  
    If i Mod 8 = 0 Then  
        sakebi = sakebi & "あは一ん。"  
    End If  
    If Len(sakebi) = 0 Then  
        sakebi = "!!"  
    End If  
    MsgBox i & sakebi  
End Sub
```

【課題 4.6】

```
Sub プレナベアツ 2()  
    Dim i As Integer, sakebi As String  
    i = Val(InputBox("1 以上 40 以下の整数を入力"))  
    sakebi = ""  
    If i Mod 3 = 0 Or i Mod 10 = 3 Or i ¥ 10 = 3 Then  
        sakebi = "あへつ。"  
    End If  
    If i Mod 8 = 0 Then  
        sakebi = sakebi & "あは一ん。"  
    End If  
    If Len(sakebi) = 0 Then  
        sakebi = "!!"  
    End If  
    MsgBox i & sakebi  
End Sub
```

5 繰り返し処理

【課題 5.1】

```
Sub 平均()  
    Dim n As Integer, i As Integer  
    Dim rng As Range, sum As Double  
    MsgBox "D2:D6 の平均"  
    Sheets("Sheet1").Select  
    Set rng = Range("D2:D6")  
    n = rng.Cells.Count  
    sum = 0  
    For i = 1 To n  
        sum = sum + rng.Cells(i).Value  
    Next i  
    MsgBox "平均=" & sum / n  
End Sub
```

【課題 5.2】

```
Sub 最大値()  
    Dim n As Integer, i As Integer  
    Dim rng As Range, m As Integer  
    MsgBox "D2:D6 の最大"  
    Sheets("Sheet1").Select  
    Set rng = Range("D2:D6")  
    n = rng.Cells.Count  
    m = 1  
    For i = 2 To n  
        If rng.Cells(m).Value < rng.Cells(i).Value Then  
            m = i  
        End If  
    Next i  
    MsgBox "最大値=" & rng.Cells(m).Value  
End Sub
```

【課題 5.3】

```
Sub 空セル横探索()  
    Dim r As Integer, c As Integer  
    MsgBox "アクティブセル直右の最初の空セルの列番号"  
    r = ActiveCell.Row  
    c = ActiveCell.Column  
    Do Until IsEmpty(Cells(r, c))  
        c = c + 1  
    Loop  
    MsgBox "最初の空セルの列番号=" & c  
End Sub
```

【課題 5.4】

```
Sub ナベアツコンプリート()  
    Dim i As Integer, sakebi As String  
    For i = 1 To 40  
        sakebi = ""  
        If i Mod 3 = 0 Or i Mod 10 = 3 Or i ¥ 10 = 3 Then  
            sakebi = "あへつ。"  
        End If  
        If i Mod 8 = 0 Then  
            sakebi = sakebi & "あは一ん。"  
        End If  
        If Len(sakebi) = 0 Then  
            sakebi = "!!"  
        End If  
        MsgBox i & sakebi  
    Next i  
End Sub
```

6 配列の活用

【課題 6.1】

```
Sub フィボナッチ数列一覧()  
    Dim fibo(20) As Long, n As Integer  
    fibo(0) = 0  
    fibo(1) = 1  
    For n = 2 To 20  
        fibo(n) = fibo(n - 1) + fibo(n - 2)  
    Next n  
    n = Val(InputBox("何番目?"))  
    MsgBox n & "番目のフィボナッチ数は" & fibo(n)  
End Sub
```

【課題 6.2】

```
Sub 平均点以上人数()  
    Dim s() As String, x() As Double  
    Dim rng As Range, n As Integer, i As Integer, k As Integer  
    Dim sum As Double  
    Set rng = Application.InputBox(prompt:= _  
        "氏名と点数の範囲は?", Type:=8)  
    n = rng.Rows.Count  
    ReDim s(n), x(n)  
    sum = 0  
    For i = 1 To n  
        s(i) = rng.Cells(i, 1).Value  
        x(i) = rng.Cells(i, 2).Value  
        sum = sum + x(i)  
    Next i  
    MsgBox "平均点=" & sum / n  
    k = 0  
    For i = 1 To n  
        If x(i) >= sum / n Then  
            k = k + 1  
        End If  
    Next i  
    MsgBox "平均点以上の人数は" & k  
End Sub
```

【課題 6.3】

```
Sub 偏差計算()  
    Dim s() As String, x() As Double  
    Dim rng As Range, n As Integer, i As Integer  
    Dim sum As Double  
    Set rng = Application.InputBox(prompt:= _  
        "氏名と点数の範囲は?", Type:=8)  
    n = rng.Rows.Count  
    ReDim s(n), x(n)  
    sum = 0  
    For i = 1 To n  
        s(i) = rng.Cells(i, 1).Value  
        x(i) = rng.Cells(i, 2).Value  
        sum = sum + x(i)  
    Next i  
    For i = 1 To n  
        rng.Cells(i, 2).Value = x(i) - sum / n  
    Next i  
End Sub
```

7 ユーザ定義型（構造体）の活用

【課題 7.1】

```
Sub 成績照会2()  
    Dim x() As Kojin  
    Dim rng As Range, n As Integer, i As Integer  
    Dim namae As String  
    Set rng = Application.InputBox(prompt:= _  
        "氏名と点数の範囲は?", Type:=8)  
    n = rng.Rows.Count  
    ReDim x(n)  
    For i = 1 To n  
        x(i).namae = rng.Cells(i, 1).Value  
        x(i).tensu = rng.Cells(i, 2).Value  
    Next i  
    namae = InputBox("氏名を入力")  
    For i = 1 To n  
        If x(i).namae = namae Then  
            MsgBox namae & "さんの成績は" & x(i).tensu  
            Exit Sub  
        End If  
    Next i  
    MsgBox namae & "という人はいません。"  
End Sub
```

【課題 7.2】

```
Sub 平均点以上2()  
    Dim x() As Kojin  
    Dim rng As Range, n As Integer, i As Integer  
    Dim sum As Double  
    Set rng = Application.InputBox(prompt:= _  
        "氏名と点数の範囲は?", Type:=8)  
    n = rng.Rows.Count  
    ReDim x(n)  
    sum = 0  
    For i = 1 To n  
        x(i).namae = rng.Cells(i, 1).Value  
        x(i).tensu = rng.Cells(i, 2).Value  
        sum = sum + x(i).tensu  
    Next i  
    MsgBox "平均点=" & sum / n  
    For i = 1 To n  
        If x(i).tensu >= sum / n Then  
            MsgBox x(i).namae & "さん、平均点以上!"  
        End If  
    Next i  
End Sub
```

【課題 7.3】

```
Sub 順位で検索()  
    Dim x() As Kojin  
    Dim rng As Range, n As Integer, i As Integer  
    Dim s As Kojin, j As Integer, m As Integer  
    Dim juni As Integer  
    Set rng = Application.InputBox(prompt:= _  
        "氏名と点数の範囲は?", Type:=8)  
    n = rng.Rows.Count  
    ReDim x(n)  
    For i = 1 To n  
        x(i).namae = rng.Cells(i, 1).Value  
        x(i).tensu = rng.Cells(i, 2).Value  
    Next i  
    For i = 1 To n - 1  
        m = i  
        For j = i + 1 To n  
            If x(m).tensu < x(j).tensu Then  
                m = j  
            End If  
        Next j  
        s = x(i): x(i) = x(m): x(m) = s  
    Next i  
    juni = Val(InputBox("第何位の人を表示?"))  
    MsgBox "第" & juni & "位の人は" & x(juni).namae  
End Sub
```

8 副プログラム

【課題 8.1】

```
Function 三乗(ByVal x As Double) As Double  
    三乗 = x ^ 3  
End Function
```

【課題 8.2】

```
Function 二進数2(ByVal n As Integer) As String  
    Dim b As String  
    b = ""  
    Do  
        b = (n Mod 2) & b  
        n = n \ 2  
    Loop Until n = 0  
    二進数2 = b  
End Function  
  
Sub 二進数変換2()  
    Dim i As Integer  
    For i = 1 To 16  
        MsgBox i & "の二進数表現は" & 二進数2(i)  
    Next i  
End Sub
```

【課題 8.3】

```
Sub 期間内日数()  
    Dim md1 As String, mm1 As Integer, dd1 As Integer  
    Dim md2 As String, mm2 As Integer, dd2 As Integer  
    Dim u As Integer, isuruu As Boolean  
    md1 = InputBox("期間初日を入力(例: 3 月 4 日なら 0304)")  
    月日数値化 md1, mm1, dd1  
    md2 = InputBox("期間終了日を入力(例: 3 月 7 日なら 0307)")  
    月日数値化 md2, mm2, dd2  
    u = Val(InputBox("うるう年(Yes なら 1, No なら 0)?"))  
    If u = 1 Then  
        isuruu = True  
    Else  
        isuruu = False  
    End If  
    MsgBox "期間内日数は" & _  
        日数(mm2, dd2, isuruu) - 日数(mm1, dd1, isuruu) + 1 & "です。"  
End Sub
```

9 コントロールの活用

《解答は教科書に記載されているので省略》

10 システム製作実践

【課題 10.1】

```
Sub 表全体選択()  
,  
,  
    ' 表全体選択 Macro  
,  
,  
    Range("A3").Select  
    Range(Selection, Selection.End(xlDown)).Select  
    Range(Selection, Selection.End(xlToRight)).Select  
End Sub
```

【課題 10.2】

```
Sub 点数全体選択()  
,  
,  
    ' 点数全体選択 Macro  
,  
,  
    Range("B4").Select  
    Range(Selection, Selection.End(xlDown)).Select  
End Sub
```

【課題 10.3】

```
Sub データ統合ソート付き ()
    Dim sh As Object, org As Range
    Dim n As Integer, k As Integer
    Sheets("A組").Select
    Range("A3:C3").Select
    Selection.Copy
    Sheets.Add
    Range("A3").Select
    ActiveSheet.Paste
    ActiveSheet.Name = "統合ソート付き"
    Range("A1").Value = "成績表（全クラス）"
    書式変更
    Set org = Sheets("統合ソート付き").Range("A3")
    n = 1
    For Each sh In Sheets
        If Right(sh.Name, 1) = "組" Then
            sh.Activate
            Range("A4").Select
            Range(Selection, Selection.End(xlDown)).Select
            Range(Selection, Selection.End(xlToRight)).Select
            Selection.Copy
            k = Selection.Rows.Count
            Sheets("統合ソート付き").Select
            org.Offset(n, 0).Select
            ActiveSheet.Paste
            n = n + k
        End If
    Next sh
    Sheets("統合ソート付き").Select
    Range("A3").Select
    Application.CutCopyMode = False
    ActiveWorkbook.Worksheets("統合ソート付き").Sort.SortFields.Clear
    点数全体選択
    ActiveWorkbook.Worksheets("統合ソート付き").Sort.SortFields.Add2 _
        Key:=Selection, SortOn:=xlSortOnValues, _
        Order:=xlDescending, DataOption:=xlSortNormal
    With ActiveWorkbook.Worksheets("統合ソート付き").Sort
        表全体選択
        .SetRange Selection
        .Header = xlYes
        .MatchCase = False
        .Orientation = xlTopToBottom
        .SortMethod = xlPinYin
        .Apply
    End With
    Range("A1").Select
End Sub
```

【課題 10.4】

```
Sub 連続印刷()  
    Dim busho As String  
    Dim num As Integer  
    For num = 1 To 5  
        If num = 1 Then  
            busho = "総務課"  
        ElseIf num = 2 Then  
            busho = "人事課"  
        ElseIf num = 3 Then  
            busho = "経理課"  
        ElseIf num = 4 Then  
            busho = "営業課"  
        ElseIf num = 5 Then  
            busho = "システム課"  
        Else  
            MsgBox "番号が間違っています。"  
            Exit Sub  
        End If  
        Range("A3").Select  
        With ActiveWorkbook.Worksheets("名簿").Sort  
            .SortFields.Clear  
            入社年全体選択  
            .SortFields.Add2 Key:=Selection, _  
            SortOn:=xlSortOnValues, Order:=xlAscending, _  
            DataOption:=xlSortNormal  
            表全体選択  
            .SetRange Selection  
            .Header = xlYes  
            .MatchCase = False  
            .Orientation = xlTopToBottom  
            .SortMethod = xlPinYin  
            .Apply  
        End With  
        Selection.AutoFilter  
        表全体選択  
        Selection.AutoFilter Field:=5, Criteria1:=busho  
        ActiveWindow.SelectedSheets.PrintPreview  
        Selection.AutoFilter  
        With ActiveWorkbook.Worksheets("名簿").Sort  
            .SortFields.Clear  
            フリガナ全体選択  
            .SortFields.Add2 Key:=Selection, _  
            SortOn:=xlSortOnValues, Order:=xlAscending, _  
            DataOption:=xlSortNormal  
            表全体選択  
            .SetRange Selection  
            .Header = xlYes  
            .MatchCase = False  
            .Orientation = xlTopToBottom  
            .SortMethod = xlPinYin  
            .Apply  
        End With  
    Next num  
End Sub
```

11 VBA による統計解析

【課題 11.1】For 文を用いて平均と分散を求めている。平均身長は野球選手が高い。分散はサッカー選手の方が大きいので、サッカー選手は身長の高い選手から低い選手までいることがわかる。

```
Sub 平均分散()  
    Cells(22, 3) = 0  
    Cells(22, 5) = 0  
    For i = 2 To 21  
        Cells(22, 3) = Cells(22, 3) + Cells(i, 3)  
        Cells(22, 5) = Cells(22, 5) + Cells(i, 5)  
    Next i  
    Cells(22, 3) = Cells(22, 3) / 20  
    Cells(22, 5) = Cells(22, 5) / 20  
    Cells(23, 3) = 0  
    Cells(23, 5) = 0  
    For i = 2 To 21  
        Cells(23, 3) = Cells(23, 3) + (Cells(i, 3) - Cells(22, 3)) ^ 2  
        Cells(23, 5) = Cells(23, 5) + (Cells(i, 5) - Cells(22, 5)) ^ 2  
    Next i  
    Cells(23, 3) = Cells(23, 3) / 20  
    Cells(23, 5) = Cells(23, 5) / 20  
End Sub
```

(追加課題) 解答は変数を使用していないので、変数を使って書き直してみなさい。

【課題 11.2】モンテカルロ法により乱数を 100 個を発生し、その点が放物線の上か下かを判断し、下の点の割合に四角形面積 16 をかけることにより面積を求める。

```
Sub 図形面積()  
    Dim nai As Integer  
    Cells(1, 1) = "回数"  
    Cells(1, 2) = "X"  
    Cells(1, 3) = "Y"  
    Cells(1, 4) = "IN/OUT"  
    Cells(1, 5) = "面積"  
    nai = 0  
    For i = 1 To 100  
        x = 4 * Rnd() - 2  
        y = Rnd()  
        Cells(i + 1, 1) = i  
        Cells(i + 1, 2) = x  
        Cells(i + 1, 3) = y  
        If (y <= 4 - x * x) Then  
            nai = nai + 1  
            Cells(i + 1, 4) = "IN"  
        End If  
        If (y > 4 - x * x) Then  
            Cells(i + 1, 4) = "OUT"  
        End If  
    Next i  
    Cells(2, 5) = (16 * nai) / 100  
End Sub
```

(追加課題) 積分を用いると正確な面積を求めることができるので、モンテカルロ法の結果と比較しなさい。

【課題 11.3】For 文を用いてじゃんけんを 5 回実施する。変数 **ten** を定義して、パーで勝つと 3、チョキで勝つと 2、グーで勝つと 1 を **ten** に加える。

```
Sub じゃんけん()  
    Cells(1, 1) = "コンの手"  
    Cells(1, 2) = "君の手"  
    Cells(1, 3) = "勝敗"  
    Cells(8, 1) = "勝点"  
    ten = 0  
    For I = 2 To 6  
        com = Int(3 * Rnd)  
        you = InputBox("君の手を入力(ぐう 0, ちょき 1, ぱあ 2)?")  
        Cells(I, 1) = com  
        Cells(I, 2) = you  
        If (you = 2 And com = 0) Then  
            Cells(I, 3) = "君の勝ち"  
            ten = ten + 3  
        End If  
        If (you = 1 And com = 2) Then  
            Cells(I, 3) = "君の勝ち"  
            ten = ten + 2  
        End If  
        If (you = 0 And com = 1) Then  
            Cells(I, 3) = "君の勝ち"  
            ten = ten + 1  
        End If  
    Next I  
    Cells(8, 2) = ten  
End Sub
```

(追加課題) じゃんけんでは負けた時は点数が減るように改良しなさい。

12 アルゴリズム

【課題 12.1】クイックソートの解答のみ記している。単純ソートはこの例を参考に作成しなさい。

```
Type Gakusei
    bango As Integer
    tensu As Double
End Type

Sub QuickSort(ByRef data() As Gakusei, ByVal left As Integer, ByVal right As Integer)
    Dim p As Integer, i As Integer, somewhere As Integer
    Dim pivot As Gakusei, tmp As Gakusei
    If left < right Then
        somewhere = (left + right) / 2
        pivot = data(somewhere)
        data(somewhere) = data(left)
        p = left
        For i = left + 1 To right
            If data(i).bango < pivot.bango Then
                p = p + 1
                tmp = data(p)
                data(p) = data(i)
                data(i) = tmp
            End If
        Next i
        data(left) = data(p)
        data(p) = pivot
        Call QuickSort(data, left, p - 1)
        Call QuickSort(data, p + 1, right)
    End If
End Sub

Sub 課題 12_1_クイックソート()
    Dim x() As Gakusei
    Dim rng As Range, n As Integer, i As Integer
    Dim saki As Range
    Set rng = Application.InputBox(prompt:= _
        "データの範囲は?", Type:=8)
    Set saki = Application.InputBox(prompt:= _
        "出力先の先頭セルは?", Type:=8)
    n = rng.Rows.Count
    ReDim x(n)
    For i = 1 To n
        x(i).bango = rng.Cells(i, 1).Value
        x(i).tensu = rng.Cells(i, 2).Value
    Next i
    Call QuickSort(x, 1, n)
    For i = 1 To n
        saki.Offset(i - 1, 0).Value = x(i).bango
        saki.Offset(i - 1, 1).Value = x(i).tensu
    Next i
End Sub
```

【課題 12.2】2 分探索の解答のみ記している。線形探索はこの例を参考に作成しなさい。

```
Type Gakusei
    bango As Integer
    tensu As Double
End Type

Function BinarySearch(ByRef data() As Gakusei, ByVal num As Integer, ByVal key As Double) _
    As Integer
    Dim pos As Integer, left As Integer, right As Integer
    Dim found As Boolean
    left = 1
    right = num
    found = False
    Do While left <= right And found = False
        pos = (left + right) / 2
        If key < data(pos).bango Then
            right = pos - 1
        ElseIf key > data(pos).bango Then
            left = pos + 1
        Else
            found = True
        End If
    Loop
    If found = False Then
        pos = -1
    End If
    BinarySearch = pos
End Function

Sub 課題 12_2_二分探索()
    Dim x() As Gakusei
    Dim rng As Range, n As Integer, i As Integer
    Dim target As Double, position As Integer
    MsgBox ("注意：二分探索では、キーのデータの並びが整列されていなければなりません。")
    Set rng = Application.InputBox(prompt:= _
        "データの範囲は?", Type:=8)
    n = rng.Rows.Count
    ReDim x(n)
    For i = 1 To n
        x(i).bango = rng.Cells(i, 1).Value
        x(i).tensu = rng.Cells(i, 2).Value
    Next i
    target = InputBox("学生番号を入力してください。")
    position = BinarySearch(x, n, target)
    If position < 0 Then
        MsgBox "お探しの学生は履修していません。"
    Else
        Range(rng.Cells(position, 1), rng.Cells(position, 2)).Select
        MsgBox "学生番号 " & x(position).bango & " の点数は " & x(position).tensu & _
            " 点です。"
    End If
End Sub
```

【課題 12.3】（クイックソートの解答）単純ソートはこの例を参考に作成しなさい。

```
Type Gakusei
    namae As String
    tensu As Double
End Type

Sub QuickSort(ByRef data() As Gakusei, ByVal left As Integer, ByVal right As Integer)
    Dim p As Integer, i As Integer, somewhere As Integer
    Dim pivot As Gakusei, tmp As Gakusei
    If left < right Then
        somewhere = (left + right) / 2
        pivot = data(somewhere)
        data(somewhere) = data(left)
        p = left
        For i = left + 1 To right
            If StrComp(data(i).namae, pivot.namae) = -1 Then
                p = p + 1
                tmp = data(p)
                data(p) = data(i)
                data(i) = tmp
            End If
        Next i
        data(left) = data(p)
        data(p) = pivot
        Call QuickSort(data, left, p - 1)
        Call QuickSort(data, p + 1, right)
    End If
End Sub

Sub 課題 12_3_クイックソート()
    Dim x() As Gakusei
    Dim rng As Range, n As Integer, i As Integer
    Dim saki As Range
    Set rng = Application.InputBox(prompt:= _
        "データの範囲は?", Type:=8)
    Set saki = Application.InputBox(prompt:= _
        "出力先の先頭セルは?", Type:=8)
    n = rng.Rows.Count
    ReDim x(n)
    For i = 1 To n
        x(i).namae = rng.Cells(i, 1).Value
        x(i).tensu = rng.Cells(i, 2).Value
    Next i
    Call QuickSort(x, 1, n)
    For i = 1 To n
        saki.Offset(i - 1, 0).Value = x(i).namae
        saki.Offset(i - 1, 1).Value = x(i).tensu
    Next i
End Sub
```

【課題 12.3】（2 分探索の解答）線形探索はこの例を参考に作成しなさい。

```
Type Gakusei
    namae As String
    tensu As Double
End Type

Function BinarySearch(ByRef data() As Gakusei, ByVal num As Integer, ByVal key As String) _
    As Integer
    Dim pos As Integer, left As Integer, right As Integer
    Dim found As Boolean
    left = 1
    right = num
    found = False
    Do While left <= right And found = False
        pos = (left + right) / 2
        If StrComp(key, data(pos).namae) = -1 Then
            right = pos - 1
        ElseIf StrComp(key, data(pos).namae) = 1 Then
            left = pos + 1
        Else
            found = True
        End If
    Loop
    If found = False Then
        pos = -1
    End If
    BinarySearch = pos
End Function

Sub 課題 12_3_二分探索()
    Dim x() As Gakusei
    Dim rng As Range, n As Integer, i As Integer
    Dim target As String, position As Integer
    MsgBox ("注意：二分探索では、キーのデータの並びが整列されていなければなりません。")
    Set rng = Application.InputBox(prompt:= _
        "データの範囲は?", Type:=8)
    n = rng.Rows.Count
    ReDim x(n)
    For i = 1 To n
        x(i).namae = rng.Cells(i, 1).Value
        x(i).tensu = rng.Cells(i, 2).Value
    Next i
    target = InputBox("学生の氏名をカタカナで入力してください。")
    position = BinarySearch(x, n, target)
    If position < 0 Then
        MsgBox "お探しの学生は履修していません。"
    Else
        Range(rng.Cells(position, 1), rng.Cells(position, 2)).Select
        MsgBox "学生： " & x(position).namae & " の点数は " & x(position).tensu & _
            " 点です。"
    End If
End Sub
```

13 データベース処理

【課題 13.1】「メニュー別」シートの A1 と A2 の日付を下図のように変更してから、マクロ「メニュー別数量」を実行すれば、下図のような結果が得られる。

	A	B	C	D	E	F	G	H
1	2015/1/1	から						
2	2015/1/31	まで						
3								
4			メニューコード	メニュー	価格	数量	売上合計	
5				1 ラーメン	600	3189	1913400	
6				2 ラーメン大盛	700	491	343700	
7				3 チャーシューメン	750	2337	1752750	
8				4 チャーシューメン大盛	850	507	430950	
9				5 ちゃんぽん	600	1342	805200	
10				6 ちゃんぽん大盛	700	293	205100	
11				7 焼きそば	600	935	561000	
12				8 焼きそば大盛	700	192	134400	
13				9 半チャーハン	400	963	385200	
14				10 餃子	400	1424	569600	
15				11 野菜炒め	400	432	172800	
16				12 おにぎり	150	915	137250	
17				13 小ライス	150	891	133650	
18								

【課題 13.2】「曜日時間帯別」シートの A1 と A2 を下図のように変更してから、

- ・ (バージョン 05 では) 2 つのマクロ「曜日時間帯別人数」と「曜日時間帯別売上合計」を、
- ・ (バージョン 06 では) マクロ「曜日時間帯別集計」を、

それぞれ実行すれば、下図のような結果が得られる。

	A	B	C	D	E	F	G	H	I	J	K	L	M	N
1	2015/1/1	から												
2	2015/1/31	まで												
3														
4			客人数	11	12	13	14	15	16	17	18	19	合計	
5			日	187	311	141	67	72	83	204	371	311	1747	
6			月	115	211	83	60	61	84	75	179	188	1056	
7			火	111	158	75	73	63	97	95	187	134	993	
8			水	99	175	80	49	46	85	91	162	196	983	
9			木	143	190	73	56	55	82	103	194	230	1126	
10			金	109	188	98	85	86	123	133	245	249	1316	
11			土	218	373	197	94	75	68	207	469	364	2065	
12			合計	982	1606	747	484	458	622	908	1807	1672	9286	
13														
14			売上合計	11	12	13	14	15	16	17	18	19	合計	
15			日	150950	250650	116400	56550	57200	69350	167200	298400	251100	1417800	
16			月	89900	171650	67050	49300	47800	66250	62800	145600	151000	851350	
17			火	89850	124700	59250	62650	51850	78650	80450	153650	108200	809250	
18			水	81550	139500	66800	39700	38450	68350	74200	135100	160000	803650	
19			木	117950	152750	58700	45500	45850	63450	82350	162450	186750	915750	
20			金	84150	150050	72950	68700	67300	100500	104850	200450	204200	1053150	
21			土	182950	304150	160500	77950	61050	55300	164850	385050	302250	1694050	
22			合計	797300	1293450	601650	400350	369500	501850	736700	1480700	1363500	7545000	
23														
24			客単価	11	12	13	14	15	16	17	18	19	合計	
25			日	807.219	805.949	825.532	844.03	794.444	835.542	819.608	804.313	807.395	811.563	
26			月	781.739	813.507	807.831	821.667	783.607	788.69	837.333	813.408	803.191	806.203	
27			火	809.459	789.241	790	858.219	823.016	810.825	846.842	821.658	807.463	814.955	
28			水	823.737	797.143	835	810.204	835.87	804.118	815.385	833.951	816.327	817.548	
29			木	824.825	803.947	804.11	812.5	833.636	773.78	799.515	837.371	811.957	813.277	
30			金	772.018	798.138	744.388	808.235	782.558	817.073	788.346	818.163	820.08	800.266	
31			土	839.22	815.416	814.721	829.255	814	813.235	796.377	821.002	830.357	820.363	
32			合計	811.914	805.386	805.422	827.169	806.769	806.833	811.344	819.424	815.49	812.513	
33														

付録 A Excel のデータベース機能

【課題 A.1】「成績データ」シートでオートフィルタを ON にし、「氏名」のプルダウンボタンから「テキストフィルタ」→「指定の値で終わる」→下図のように指定。

オートフィルタ オプション

抽出条件の指定：

氏名

あかり

で終わる

☐ AND(A) ☒ OR(Q)

ひかり

で終わる

? を使って、任意の 1 文字を表すことができます。
* を使って、任意の文字列を表すことができます。

OK キャンセル

結果は下図の通り（7 人が該当）。

	A	B	C	D	E	F	G	H	I	J
1	学籍番号	クラス	性別	氏名	簿記原簿	商学入	経営学入	会計学入	経済学入	5科目平均
81	215CV30	V	女	五十嵐 あかり	84	71	56	81	83	75
119	215CW18	W	女	小松 ひかり	84	81	72	83	77	79.4
125	215CW24	W	女	高山 ひかり	81	60	73	93	84	78.2
182	215CX31	X	女	金子 あかり	60	65	79	61	54	63.8
238	215CY37	Y	女	中村 ひかり	70	66	60	75	71	68.4
254	215CZ03	Z	女	島田 ひかり	70	79	77	70	59	71
267	215CZ16	Z	女	古川 あかり	63	79	71	74	54	68.2
302										

【課題 A.2】「成績データ」シートのオートフィルタが ON の状態で、「5 科目平均」のプルダウンボタンから「数値フィルタ」→「指定の範囲内」→下図のように指定。

オートフィルタ オプション

抽出条件の指定：

5科目平均

70

以上

☒ AND(A) ☐ OR(Q)

80

より小さい

? を使って、任意の 1 文字を表すことができます。
* を使って、任意の文字列を表すことができます。

要変更

OK キャンセル

結果は下図の通り（130 人が該当）。

	A	B	C	D	E	F	G	H	I	J
1	学籍番号	クラス	性別	氏名	簿記原簿	商学入	経営学入	会計学入	経済学入	5科目平均
6	215CU05	U	男	川崎 大雅	81	64	71	88	84	77.6
9	215CU08	U	男	早川 樹	74	81	71	75	75	75.2
10	215CU09	U	男	久保 悠人	77	70	59	75	95	75.2
11	215CU10	U	男	藤田 大和	73	74	64	74	68	70.6
13	215CU12	U	男	荒木 悠	77	77	64	75	70	72.6
14	215CU13	U	男	五十嵐 晴翔	81	67	82	75	76	76.2

準備完了 300 レコード中 130 個が見つかりました

【課題 A.3】 N8 セルに =AVERAGEIF(\$C\$2:\$C\$301, \$L8, E\$2:E\$301) と入力し、それを残りのセルにコピー・数式のみ貼り付けすれば、下図のように得られる。

	L	M	N	O	P	Q	R	S	T
1		人数	簿記原理	商学入門	経営学入門	会計学入門	経済学入門	5科目平均	
2	U	50	71.56	71.6	67.56	74.8	67.64	70.632	
3	V	50	69.98	70.72	67.98	73.94	63.4	69.204	
4	W	50	71.26	69.8	67.8	75.98	67.94	70.556	
5	X	50	70.32	71	68.6	74.3	65.38	69.92	
6	Y	50	69.84	70.22	67.8	74.18	65.36	69.48	
7	Z	50	68.96	72.7	67.48	73.76	61.76	68.932	
8	女	100	71.41	70.79	67.83	75.82	66.91	70.552	
9	男	200	69.775	71.115	67.89	73.83	64.415	69.405	
10									

【課題 A.4】 M12 セルに =COUNTIFS(\$B\$2:\$B\$301, \$L12, \$C\$2:\$C\$301, M\$11) と入力し、それを残りのセルにコピー・数式のみ貼り付けすれば、下図のように得られる。

	L	M	N	O
11	人数	女	男	
12	U	15	35	
13	V	10	40	
14	W	18	32	
15	X	13	37	
16	Y	22	28	
17	Z	22	28	
18				

【課題 A.5】 M20 セルに =AVERAGEIFS(\$J\$2:\$J\$301, \$B\$2:\$B\$301, \$L20, \$C\$2:\$C\$301, M\$19) と入力し、それを残りのセルにコピー・数式のみ貼り付けすれば、下図のように得られる。

	L	M	N	O
19	5科目平均	女	男	
20	U	70.72	70.59429	
21	V	69.78	69.06	
22	W	71.9	69.8	
23	X	71.78462	69.26486	
24	Y	69.05455	69.81429	
25	Z	70.45455	67.73571	
26				

【課題 A.6】「ピボット 1」シートで値フィールドの「合計/売上合計」を削除し、代わりにフィールドリストから「数量」をセットすれば、下図のように得られる。

	A	B	C	D
1				
2				
3	メニューコード	メニュー	合計 / 数量	
4		① ラーメン	9311	
5		② ラーメン大盛	1396	
6		③ チャーシューメン	6736	
7		④ チャーシューメン大盛	1388	
8		⑤ ちゃんぽん	3975	
9		⑥ ちゃんぽん大盛	782	
10		⑦ 焼きそば	2768	
11		⑧ 焼きそば大盛	527	
12		⑨ 半チャーハン	2724	
13		⑩ 餃子	4056	
14		⑪ 野菜炒め	1351	
15		⑫ おにぎり	2667	
16		⑬ 小ライス	2604	
17	総計		40285	
18				

【課題 A.7】「内訳」シートの H1 に「時間帯」と入力し、H2 に

=VLOOKUP(B2, 伝票履歴!\$A\$2:\$F\$13586, 3, FALSE)

と入力して残りのセルにコピー&貼り付けを行う。A1 セル選択後、新規ワークシートにピボットテーブルを追加して、シート名を「ピボット 3」とする。フィールドリストから「メニューコード」を行フィールドに、「時間帯」を列フィールドに、「売上合計」を値フィールドにそれぞれセットし、実習 A.20 の手順で「メニューコード」と「メニュー」の併記をする。その結果は下図のように得られる。

	A	B	C	D	E	F	G	H	I	J	K	L	M
1													
2													
3	合計 / 売上合計		時間帯										
4	メニューコード	メニュー	11	12	13	14	15	16	17	18	19	総計	
5		① ラーメン	588000	998400	442200	279600	282600	399600	536400	1068000	991800	6586600	
6		② ラーメン大盛	95900	163100	81900	51100	42700	65900	114100	172200	190400	977200	
7		③ チャーシューメン	537000	831750	404250	255750	233250	343500	520500	993750	932250	5052000	
8		④ チャーシューメン大盛	122400	204000	102850	59500	53550	86700	102000	222700	226100	1179900	
9		⑤ ちゃんぽん	232200	402000	195600	130200	130200	164400	230400	444600	455400	2385000	
10		⑥ ちゃんぽん大盛	67900	100800	43400	32200	22400	38500	50400	98000	93800	547400	
11		⑦ 焼きそば	170400	281400	130200	80400	78600	132600	172200	316200	298800	1660800	
12		⑧ 焼きそば大盛	38500	63700	28700	23800	19600	21000	40600	66500	66500	368900	
13		⑨ 半チャーハン	108000	189600	84400	48400	52800	72800	99600	226000	208000	1089600	
14		⑩ 餃子	170800	277600	123600	80400	74800	112000	169600	315200	298400	1622400	
15		⑪ 野菜炒め	60400	86800	39200	30000	27200	44000	54000	101600	97200	540400	
16		⑫ おにぎり	43500	69600	32100	22650	19050	28050	36900	75900	72300	400050	
17		⑬ 小ライス	40050	67200	31500	21000	20250	25950	37200	74250	73200	390600	
18	総計		2275050	3735950	1739900	1115000	1057000	1534900	2163900	4174900	4004150	21800750	
19													